



DE LA RECHERCHE À L'INDUSTRIE



LA PROCÉDURE PASAPAS ET LES PROCÉDURES UTILISATEURS

DISPONIBLE SUR : [HTTP://WWW-CAST3M.CEA.FR/INDEX.PHP?XML=FORMATIONS](http://www-cast3m.cea.fr/index.php?xml=formations)



www.cea.fr

François DI PAOLA, Caroline GUERIN


DERNIÈRE MODIFICATION : 30 NOVEMBRE 2022

- **Présentation de Cast3M et rappels sur PASAPAS**
- **Fonctionnement de PASAPAS**
- **Résolution de l'équilibre mécanique : UNPAS**
 - Exercice 1 : force suiveuse
 - Exercice 2 : rupture par suppression d'éléments
- **Résolution de l'équilibre thermique : TRANSNON**
 - Exercice 3 : source de chaleur variable
 - Exercice 4 : contact thermo-mécanique

PRÉSENTATION DE CAST3M
RAPPELS SUR LA PROCÉDURE PASAPAS

Logiciel de simulation utilisant la méthode des **éléments finis** en **mécanique/thermique** des **structures** et des **fluides**


- Résolution d'**équations aux dérivées partielles** par la méthode des éléments finis
- **Systeme complet** : solveur, pré/post-processeur, visualisation, import/export des données ...




- Basé sur un langage de commande : **Gibiane** (orienté objet)

DE NOMBREUX DOMAINES D'APPLICATION


- **Mécanique des structures (historique)**
Quasi-statique (non linéarités matériau, géométrie, conditions limites)
Contact/frottement, **Flambage**
Dynamique (temporelle, modale, interaction fluide structure)
Rupture (XFEM, propagation dynamique, zones cohésives, ...)



- **Thermique**
 Conduction, convection, rayonnement



- **Fluides**




- **Métallurgie**


- **Magnétostatique**

- **Diffusion multi espèces (**

- **Couplage thermo-hygro-mécanique**



SCAL



AMPLITUDE DEFORMEE

Temperature au temps 0.00000E+00 (0 s)

CAST3M FEKIT
 AM
 DE

Proportions de BAIN au temps : 0.00000E+00

■ Objectif

résolution de problèmes non linéaires **évolutifs** de manière incrémentale en **thermique** et en **mécanique**

le "temps" peut être physique (ex : thermique transitoire) ou non (ex : plasticité avec chargement progressif)

→ on parle donc volontiers de **variable d'évolution**

■ Types de non linéarités traitées

comportement (plasticité, endommagement, matériaux variables, ...)

géométrie (grands déplacements)

déformations (grandes rotations)

conditions limites (rayonnement, frottement, pression suiveuse, ...)

■ Créer une table contenant toutes les données du problème :

```
TAB1                = TABL ;
TAB1 . MODELE       = MOD1 ET MOD2 ;
TAB1 . CARACTERISTIQUES = MAT1 ET MAT2 ;
TAB1 . BLOCAGES_MECAIQUES = BLO1 ;
TAB1 . CHARGEMENT   = CHA1 ET CHA2 ET CHA3 ;
TAB1 . TEMPS_CALCULES = PROG 0.1 'PAS' 0.1 50. ;
TAB1 . TEMPS_SAUVES  = PROG 4. 8. 15. 16. 23. 42. ;
TAB1 . PRECISION     = 1.E-6 ;
TAB1 . GRANDS_DEPLACEMENTS = VRAI ;
...
```

certaines indices sont obligatoires, d'autres non

■ Appel à la procédure proprement dite :

```
PASAPAS TAB1 ;
```

■ Post-traitement des résultats

APERÇU DES PARAMÈTRES D'ENTRÉE

■ Généralités

MODELE (MMODEL)

CARACTERISTIQUES (MCHAML)

CHARGEMENT (CHARGEME)

Équations à résoudre, formulation E.F. (MODE)

Paramètres matériau et/ou géométriques (MATE)

Évolution des C.L. au cours du calcul (CHAR)

■ Thermique

BLOCAGES_THERMIQUES (RIGIDITE)

CELSIUS (LOGIQUE)

TEMPERATURES . 0 (CHPOINT)

Matrice de blocage des C.L. de type DIRICHLET (BLOQ)

=VRAI si les températures sont en degrés CELSIUS

Conditions initiales

■ Mécanique

BLOCAGES_MECANIQUES (RIGIDITE)

GRANDS_DEPLACEMENTS (LOGIQUE)

DEPLACEMENTS . 0 (CHPOINT)

CONTRAINTES . 0 (MCHAML)

VARIABLES_INTERNES . 0 (MCHAML)

DEFORMATIONS_INELASTIQUES . 0 (MCHAML)

Matrice de blocage des C.L. de type DIRICHLET (BLOQ)

Équilibre vérifié sur les configurations déformées

Conditions initiales

APERÇU DES PARAMÈTRES D'ENTRÉE

■ Mécanique (dynamique)

DYNAMIQUE (LOGIQUE)

=VRAI si calcul dynamique

AMORTISSEMENT (RIGIDITE)

Matrice d'amortissement

VITESSES . 0 (CHPOINT)

ACCELERATIONS . 0 (CHPOINT)

Conditions initiales

■ Instants de calcul et sauvegarde

TEMPS_CALCULES (LISTREEL)

Liste des instants de calcul (variable d'évolution)

TEMPS_SAUVES (LISTREEL)

Instants de calcul conservés dans la table à la sortie

OPTI 'SAUV' 'mon_fichier' ;

TEMPS_SAUVEGARDES (LISTREEL)

*Instants de calcul où PASAPAS fera appel à **SAUV***

MES_SAUVEGARDES (TABLE)

Grandeurs à sauvegarder (déformations totales, ...)

■ Les résultats sont rangés dans la table

TEMPS (TABLE)

Instants de calcul, correspond aux « TEMPS_SAUVES »

TEMPERATURES (TABLE)

PROPORTIONS_PHASE (TABLE)

DEPLACEMENTS (TABLE)

CONTRAINTES (TABLE)

DEFORMATIONS_INELASTIQUES (TABLE)

VARIABLES_INTERNES (TABLE)

REACTIONS (TABLE)

VITESSES (TABLE)

ACCELERATIONS (TABLE)

Champs solution calculés pour chaque « TEMPS_SAUVES »

POST TRAITEMENT (EXEMPLES)

- **Extraction des champs solution :**

à partir de l'indice dans la table

SIG1 = TAB1 . CONTRAINTE5 . 5 ;

ou bien en connaissant l'instant de calcul

SIG1 = PECHE TAB1 'CONTRAINTE5' 28.3 ;

- **Tracé en mode graphique interactif (limité) :**

EXPLORER TAB1 ;


- **Évolution temporelle d'un champ calculé :**

EV1 = EVOL 'TEMP' TAB1 'DEPLACEMENTS' 'UX' P1 ;

FONCTIONNEMENT DE PASAPAS

■ Algorithme principal

Initialisations



Fin


+ Appels à des procédures utilisateur (optionnel)

PERS01 PERS02 REEV_MEC REEV_THE
 CHARMECA CHARTHER PARATHER

ces procédures sont à définir par l'utilisateur !



FONCTIONNEMENT DE PASAPAS



■ TAB1 . 'ESTIMATION'

contient les résultats calculés/convergés par TRANSNON et UNPAS mais pas convergées au sens de **BO_BOTH**

```
TAB1 . 'ESTIMATION' . 'TEMPS'  
TAB1 . 'ESTIMATION' . 'DEPLACEMENTS'  
TAB1 . 'ESTIMATION' . 'CONTRAINTES'  
TAB1 . 'ESTIMATION' . 'TEMPERATURES'  
...  
...
```

■ TAB1 . 'CONTINUATION'

contient les résultats convergés (au sens de **BO_BOTH**) cet indice est mis à jour **à la fin du pas de temps !!!**

➔ utile pour une reprise de PASAPAS

```
TAB1 . 'CONTINUATION' . 'TEMPS'  
TAB1 . 'CONTINUATION' . 'DEPLACEMENTS'  
TAB1 . 'CONTINUATION' . 'CONTRAINTES'  
TAB1 . 'CONTINUATION' . 'TEMPERATURES'  
...  
...
```

■ TAB1 . 'WTABLE'

Variables utiles à PASAPAS (*options choisies, modèles, matériaux/chargements instanciés, résultats intermédiaires, ...*)

Quelques indices :

WTABLE . 'CHARGEMENT'	<i>Chargement utilisé pour le pas de temps</i>
WTABLE . 'THER_COURANT'	<i>Température à la dernière itération (au cours d'un pas)</i>
WTABLE . 'BLOCAGES_MECAIQUES'	<i>Matrice de blocage mécanique</i>
WTABLE . 'BLOCAGES_THERMIQUES'	<i>Matrice de blocage thermique</i>
WTABLE . 'FOR'	<i>Configuration au début du pas</i>
WTABLE . 'FOR0'	<i>Configuration initiale</i>
WTABLE . 'MODELE'	<i>Modèles</i>
WTABLE . 'CARACTERISTIQUES'	<i>Champ de caractéristiques matériau</i>

...

Plus d'infos, voir :


→ Commentaires procédure PAS_DEFA (http://www-cast3m.cea.fr/index.php?page=procedures&procedure=pas_defa)

→ Documentation PASAPAS (http://www-cast3m.cea.fr/html/Documentation_Cast3M/Pasapas.pdf)

**RÉSOLUTION DE L'ÉQUILIBRE
MÉCANIQUE
PROCÉDURE UNPAS**

■ Équations locales du problème mécanique statique

$$\begin{array}{lll}
 \text{équilibre} & \text{div}(\boldsymbol{\sigma}) + \mathbf{f} = \mathbf{0} & \text{sur } V \\
 \text{efforts imposés} & \boldsymbol{\sigma} \cdot \mathbf{n} = \mathbf{t} & \text{sur } \partial V^t \\
 \text{déplacements imposés} & \mathbf{u} = \mathbf{d} & \text{sur } \partial V^d
 \end{array}$$



■ Formulation faible + discrétisation EF

$$\begin{aligned}
 & \{F\} - \int_V [B]^T \{\sigma\} dV = \{0\} \\
 & \underbrace{\int_{\partial V^t} [N]^T \{t\} dS}_{\{F\}^S} + \underbrace{\int_{\partial V^d} [N]^T \{\sigma \cdot n\} dS}_{\{F\}^R} + \underbrace{\int_V [N]^T \{f\} dV}_{\{F\}^V} - \underbrace{\int_V [B]^T \{\sigma\} dV}_{[B]\{\sigma\}} = \{0\}
 \end{aligned}$$

Vecteurs des Forces nodales équivalentes (N)

- $\{F\}^S$ de la densité surfacique d'efforts imposés t (Pa) (PRES, FSUR, FORC, ...)
- $\{F\}^R$ de la densité surfacique d'efforts de réaction aux dépl. imposés d (m) (REAC)
- $\{F\}^V$ de la densité volumique d'efforts imposés f (N.m⁻³) (CNEQ)
- $[B]\{\sigma\}$ de la densité volumique d'efforts intérieurs (BSIG)

Matrices

- $[N]$ matrice des fonctions de forme (d'interpolation)
- $[B]$ matrice des dérivées des fonctions de forme

■ Résidu = mesure du déséquilibre

$$\{R\} = \underbrace{\{F\}^S + \{F\}^R + \{F\}^V}_{\{F\}^{ext}} - \underbrace{[B]\{\sigma\}}_{\{F\}^{int}}$$

■ Équilibre atteint lorsque

$$\|R\| < \varepsilon \cdot F^{ref}$$

$\|R\|$ norme du résidu, par exemple la norme infinie

ε précision du calcul (fournie par utilisateur)

F^{ref} effort de référence du problème considéré

■ Introduction de la matrice de rigidité élastique HPP

$$\boldsymbol{\sigma} = \mathbb{E} : \boldsymbol{\varepsilon}^{lin} + \boldsymbol{\sigma}^{nl} \quad \text{avec } \boldsymbol{\varepsilon}^{lin} = \frac{1}{2} (\nabla u + \nabla^T u)$$

$$\{\boldsymbol{\varepsilon}\}^{lin} = [B]\{U\}$$

■ L'équilibre s'écrit alors

$$[B]\{\boldsymbol{\sigma}\} = \{F\}^S + \{F\}^R + \{F\}^V$$

$$[B][E]\{\boldsymbol{\varepsilon}\}^{lin} = \{F\}^S + \{F\}^R + \{F\}^V - [B]\{\boldsymbol{\sigma}\}^{nl}$$

$$\underbrace{\int_V [B]^T [E] [B] dV}_{[K]} \{U\} = \{F\}^S + \{F\}^R + \{F\}^V - [B]\{\boldsymbol{\sigma}\}^{nl}$$

$$[K]\{U\} = \{F\}^S + \{F\}^R + \{F\}^V - [B]\{\boldsymbol{\sigma}\}^{nl}$$

$\{U\}$ déplacements nodaux

$\{\boldsymbol{\varepsilon}\}^{lin}$ déformations linéaires (« petites » déformations)

$\{\boldsymbol{\sigma}\}^{nl}$ contraintes « complémentaires » non linéaires

$[E]$ matrice de Hooke (ELAS)

$[K]$ matrice de rigidité élastique (RIGI)

$$u(x) = [N(x)]\{U\}$$

$$\{\boldsymbol{\varepsilon}\}^{lin} = [B]\{U\} \quad \text{(EPSI)}$$

■ Matrice de blocage

$$u = d \quad \text{sur } \partial V^d$$

$$[A]\{U\} = \{d\}$$

■ Multiplicateurs de Lagrange

Pour un système linéaire sans blocages, d'inconnue $\{U\}$: $[K]\{U\} = \{F\}^S + \{F\}^V - [B]\{\sigma\}^{nl}$

On ajoute les inconnues $\{\lambda\}$

$$[K]\{U\} + \underbrace{[A]^T\{\lambda\}}_{-\{F\}^R} = \{F\}^S + \{F\}^V - [B]\{\sigma\}^{nl} \quad \text{avec } [A]\{U\} = \{d\}$$

L'équilibre s'écrit finalement :

$$\underbrace{\begin{bmatrix} [K] & A^T \\ A & 0 \end{bmatrix}}_{[\hat{K}]} \begin{Bmatrix} \{U\} \\ \{\lambda\} \end{Bmatrix} = \begin{Bmatrix} \{F\}^S + \{F\}^V - [B]\{\sigma\}^{nl} \\ d \end{Bmatrix}$$

dépendant de $\{U\}$!!
comportement non linéaire ;
grands déplacements ;
forces suiveuses ;
...

- $[A]$ matrice des blocages/rerelations (BLOQ, RELA)
- $\{d\}$ déplacements nodaux imposés (DEPI)
- $\{\lambda\}$ multiplicateurs de Lagrange

■ Calcul d'un pas de temps entre t_0 et t_1

$[t_0, \{U\}_0, \{\lambda\}_0, \{\sigma\}_0]$ état connu début du pas

$[t_1, \{U\}_1, \{\lambda\}_1, \{\sigma\}_1]$ état recherché fin du pas

■ Décomposition incrémentale des déplacements nodaux

$$\{U\}_1^i = \{U\}_0 + \Delta\{U\}_1^i \quad \text{et} \quad \Delta\{U\}_1^{i+1} = \Delta\{U\}_1^i + \delta\{U\}_1^{i+1}$$

et donc

$$\{U\}_1^{i+1} = \{U\}_1^i + \delta\{U\}_1^{i+1}$$

$\{U\}_1^i$ estimation du déplacement $\{U\}_1$ à l'itération i

$\Delta\{U\}_1^i$ estimation de l'incrément de déplacement à l'itération i

$\delta\{U\}_1^{i+1}$ correction de l'incrément de déplacement à l'itération i

■ Décomposition incrémentale des conditions sur les déplacements

$$[A]\{U\}_1^{i+1} = \{d\}_1$$

$$[A]\delta\{U\}_1^{i+1} = \{d\}_1 - [A]\{U\}_1^i$$

■ Décomposition incrémentale de l'équilibre

$$[K]\{U\}_1^{i+1} + [A]^T\{\lambda\}_1^{i+1} = \{F\}_1^S + \{F\}_1^V - [B]\{\sigma\}_1^{nl,i}$$

$$[K]\delta\{U\}_1^{i+1} = \{F\}_1^S + \{F\}_1^V - [A]^T\{\lambda\}_1^{i+1} - \underbrace{([K]\{U\}_1^i + [B]\{\sigma\}_1^{nl,i})}_{[B]\{\sigma\}_1^i}$$

$$[K]\delta\{U\}_1^{i+1} = \underbrace{\{F\}_1^S + \{F\}_1^V - [A]^T\{\lambda\}_1^{i+1} - [B]\{\sigma\}_1^i}_{\{R\}_1^i}$$

■ L'équilibre s'écrit finalement

$$\underbrace{\begin{bmatrix} K & A^T \\ A & 0 \end{bmatrix}}_{[\hat{K}]} \begin{Bmatrix} \delta U_1^{i+1} \\ \lambda_1^{i+1} \end{Bmatrix} = \begin{Bmatrix} F_1^S + F_1^V - [B]\{\sigma\}_1^i \\ \{d\}_1 - [A]\{U\}_1^i \end{Bmatrix}$$

■ Algorithme simple

$$[\{U\}_1^{i=0}, \{\lambda\}_1^{i=0}, \{\sigma\}_1^{i=0}] = [\{U\}_0, \{\lambda\}_0, \{\sigma\}_0]$$

$$\{F\}_1^{R,i=0} = -[A]^T \{\lambda\}_1^{i=0}$$

$$F^{ref} = \|\{F\}_1^S + \{F\}_1^V + \{F\}_1^{R,i=0}\|$$

$$\{R\}_1^{i=0} = \{F\}_1^S + \{F\}_1^V + \{F\}_1^{R,i=0} - [B]\{\sigma\}_1^{i=0}$$

initialisations

efforts de réactions (REAC)

norme de convergence (MAXI 'ABS')

premier résidu (BSIG)

Tant que : $\|\{R\}_1^i\| / F^{ref} \geq \varepsilon$

$$[\delta\{U\}_1^{i+1}, \{\lambda\}_1^{i+1}] = [\hat{K}]^{-1} \{R\}_1^i$$

$$\{U\}_1^{i+1} = \{U\}_1^i + \delta\{U\}_1^{i+1}$$

$$\{\varepsilon\}_1^{i+1} = \mathcal{D}(\{U\}_1^{i+1})$$

$$\{\sigma\}_1^{i+1} = \mathcal{C}(\Delta\{\varepsilon\}_1^{i+1})$$

$$\{F\}_1^{R,i+1} = -[A]^T \{\lambda\}_1^{i+1}$$

$$\{R\}_1^{i+1} = \{F\}_1^S + \{F\}_1^V + \{F\}_1^{R,i+1} - [B]\{\sigma\}_1^{i+1}$$

$$i = i + 1$$

résolution (RESO)


estim. déplacements

estim. déformations (EPSI)

estim. contraintes (COMP)


estim. réactions (REAC)

nouveau résidu (BSIG)



Fin

FONCTIONNEMENT DE UNPAS



LES PROCÉDURES UTILISATEUR DANS PASAPAS

PERSO1 REEV_MEC CHARMECA

PERSO2 REEV_THE CHARTHER PARATHER

LES PROCÉDURES UTILISATEURS : MODE D'EMPLOI

- Il existe 7 procédures utilisateurs, appelées à différentes étapes de l'algorithme de PASAPAS

PERS01	PERS02	REEV_MEC	REEV_THE
CHARMECA	CHARTHER	PARATHER	

- 1) **Choisir la procédure** à utiliser selon son emplacement dans l'algorithme et l'action désirée
par exemple : puisque PERS01 est appelée après le calcul d'un pas de temps, elle se destine à modifier le prochain pas e.g. conditions aux limites, paramètres matériau, ...

- 2) **Observer la syntaxe** attendue de la procédure
 - dans la notice de PASAPAS, ou bien
 - dans la procédure PASAPAS, UNPAS ou TRANSNON

le tableau ci-après rappelle les syntaxes des procédures

■ 3) Définir la procédure souhaitée

```
DEBP PERSO1 T1*'TABLE' ;  
  MESS 'Bonjour ! La procedure PERSO1 vous parle.' ;  
FINP ;
```

■ 4) Demander son utilisation lors de l'appel à PASAPAS

```
TAB1 . 'MODELE' = MOD1 ;  
TAB1 . 'CARACTERISTIQUES' = MAT1 ;  
TAB1 . 'TEMPS_CALCULES' = PROG 1. 2. 3. 4. ;  
TAB1 . 'PROCEDURE_PERSO1' = VRAI ;  
...  
...  
PASAPAS TAB1 ;
```

LISTE DES PROCÉDURES UTILISATEURS

Nom	Indice à activer dans la table de PASAPAS	Syntaxe	Rôle
PERSO1	TAB1 . 'PROCEDURE_PERSO1' = VRAI ;	PERSO1 T1 ;	Mise à jour du problème après calcul du pas mécanique
REEV_MEC	TAB1 . 'PROCEDURE_REEV_MEC' = VRAI ;	REEV_MEC T1 N1 ;	Idem (mais dans BO_BOTH) → le pas peut être recommencé
CHARMECA	TAB1 . 'PROCEDURE_CHARMECA' = VRAI ;	T2 = CHARMECA T1 TPS1 ;	Ajout de chargements mécaniques au cours des itérations de UNPAS
PERSO2	TAB1 . 'PROCEDURE_PERSO2' = VRAI ;	PERSO2 T1 ;	Mise à jour du problème après calcul du pas thermique
REEV_THE	TAB1 . 'PROCEDURE_REEV_THE' = VRAI ;	REEV_THE T1 N1 ;	Idem (mais dans PASAPAS) → le pas peut être recommencé
CHARTHER	TAB1 . 'PROCEDURE_CHARTHER' = VRAI ;	T2 = CHARTHER T1 TPS1 ;	Ajout de chargements thermiques au cours des itérations de TRANSNON
PARATHER	TAB1 . 'PROCEDURE_PARATHER' = VRAI ;	PARATHER T1 TPS1 ;	Mise à jour des variables externes des caractéristiques thermiques

Avec

T1 : la table de PASAPAS

TPS1 : l'instant de calcul courant

N1 : le numéro d'appel de la procédure (0 ou 1) → voir la procédure PASAPAS

T2 : la table de sortie (seulement pour les procédures CHARMECA et CHARTHER)

QUELQUES REMARQUES

- Les instructions dans ces procédures sont libres !!!

- **CHARMECA** et **CHARTHER** doivent sortir une table, avec 2 indices possibles :
 - **'ADDI_MATRICE'** qui contient les matrices de rigidité (objet de type RIGIDITE) que l'on veut ajouter au 1^{er} membre

 - **'ADDI_SECOND'** qui contient les CHPOINT que l'on veut ajouter au 2nd membre (forces nodales)

- En grands déplacements (option **'GRANDS_DEPLACEMENTS'**), **CHARMECA** est appelée sur la configuration déformée

■ On fait varier aléatoirement la carte du module de Young à chaque pas de temps

```

OPTI 'DIME' 2 'ELEM' 'QUA4' ;
P0 = 0. 0. ; P1 = 1. 0. ;
L1 = DROI 1 P0 (0. 0.1) ;
S1 = L1 TRAN 10 (1. 0.) ;
L2 = S1 COTE 3 ; P2 = L2 POIN 'PROC' (1. 0.) ;

MO = MODE S1 'MECANIQUE' ;
MA = MATE MO 'YOUN' 200.E9 'NU' 0.3 ;

BL = (BLOQ 'UX' L1) ET (BLOQ 'UY' P0) ;
F = PRES 'MASS' MO L2 -1.E8 ;
EV = EVOL 'MANU' (PROG 0. 1.) (PROG 0. 1.) ;
CHA = CHAR 'MECA' F EV ;

TAB1 = TABL ;
TAB1 . 'MODELE' = MO ;
TAB1 . 'CARACTERISTIQUES' = MA ;
TAB1 . 'BLOCAGES_MECANIQUES' = BL ;
TAB1 . 'CHARGEMENT' = CHA ;
TAB1 . 'TEMPS_CALCULES' = PROG 0.1 'PAS' 0.1 1. ;
TAB1 . 'PROCEDURE_PERSO1' = VRAI ;

```

```
PASAPAS TAB1 ;
```

```


EVU = EVOL 'TEMP' TAB1 'DEPLACEMENTS' 'UX' P2 ;
DESS EVU ;

```

```

DEBP PERSO1 T1*'TABLE' ;
MOD = T1 . 'WTABLE' . 'MO_TOT' ;
MAIL = EXTR MOD 'MAIL' ;
CHPYO = BRUI 'BLAN' 'UNIF' 200.E9 150.E9 MAIL ;
CHMYO = CHAN 'CHAM' CHPYO MOD 'RIGIDITE' ;
MAT = MATE MOD 'YOUN' CHMYO 'NU' 0.3 ;
T1 . 'WTABLE' . 'CARACTERISTIQUES' = MAT ;
FINP ;

```



EXERCICE 1 :

POUTRE AVEC FORCE SUIVEUSE

TÉLÉCHARGEZ LE FICHIER INITIAL SUR LE SITE WEB :

[HTTP://WWW-CAST3M.CEA.FR/INDEX.PHP?PAGE=EXEMPLES&EXEMPLE=FORMATION_PASAPAS_1_INITIAL](http://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_1_initial)



EXERCICE 1 : POUTRE AVEC FORCE SUIVEUSE

■ Poutre en flexion

base encastrée

effort tranchant ponctuel au sommet (perpendiculaire à la poutre)

déplacement "important" du sommet



■ **Problème : l'effort est calculé sur la configuration initiale et n'est pas mis à jour**

■ **Objectif : réappliquer l'effort correctement au cours des pas**

À vous de jouer !

Quelques indices ...

■ Objets utiles

P2 : point au sommet de la poutre, où est appliqué la force
EV1 : évolution de l'amplitude de la force à appliquer vs. temps

■ Opérateurs utiles

EXTR, COS, SIN

IPOL : pour interpoler l'amplitude de la force à l'instant de calcul

FORC : pour appliquer une force ponctuelle

EXERCICE 1 : POUTRE AVEC FORCE SUIVEUSE

■ Solution



- utiliser l'option '**GRANDS_DEPLACEMENTS**' (équilibre vérifié sur la configuration déformée)
- utiliser la procédure **PERSO1** (1 appel / pas de temps) pour re-calculer l'effort sur la configuration déformée (au début du pas)
- en faire un chargement
- écraser le chargement dans **WTABLE**

```
DEBP PERSO1 T1*'TABLE' ;
* donnees utiles
TAM = T1 . 'AMOI' ;
PF = TAM . 'POINT' ;
EV = TAM . 'EVOL' ;
* angle de rotation de la section droite
U1 = T1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
RS = EXTR U1 'RZ' PF ;
RSD = RS * 180. / PI ;
* direction de l'effort
DIR1 = (COS RSD) (SIN RSD) ;
* creation du second membre
F1 = FORC DIR1 PF ;
CHAR2 = CHAR 'MECA' F1 EV ;
T1 . 'WTABLE' . 'CHARGEMENT' = CHAR2 ;
FINP ;
```

```
TAB1 . 'GRANDS_DEPLACEMENTS' = VRAI ;
TAB1 . 'PREDICTEUR'           = 'HPP' ;
TAB1 . 'PROCEDURE_PERSO1'    = VRAI ;
TAB1 . 'AMOI'                 = TABL ;
TAB1 . 'AMOI' . 'EVOL'       = EV1 ;
TAB1 . 'AMOI' . 'POINT'     = P2 ;
```

■ Résultats

Déformée + réaction



Deplacements vs Temps

EXERCICE 1 : POUTRE AVEC FORCE SUIVEUSE

■ Solution (bis)

- idem mais avec la procédure **CHARMECA**
- supprimer le chargement mécanique initial
- pas besoin d'objet de type CHARGEMENT
- plus long : 1 appel / itération / pas de temps
- résultats identiques

```
DEBP CHARMECA T1*'TABLE' TPS1*'FLOTTANT' ;
* donnees utiles
  TAM = T1 . 'AMOI' ;
  PF = TAM . 'POINT' ;
  EV = TAM . 'EVOL' ;
* angle de rotation de la section droite
  U1 = T1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
  RS = EXTR U1 'RZ' PF ;
  RSD = RS * 180. / PI ;
* direction de l'effort
  DIR1 = (COS RSD) (SIN RSD) ;
* creation du second membre
  VAL1 = IPOL TPS1 EV ;
  F1 = FORC (VAL1 * DIR1) PF ;
  T2 = TABLE ;
  T2 . 'ADDI_SECOND' = F1 ;
FINP T2 ;
```

Procédure **CHARMECA**

```
***TAB1 . 'CHARGEMENT' = CHAR1 ;
TAB1 . 'GRANDS_DEPLACEMENTS' = VRAI ;
TAB1 . 'PREDICTEUR' = 'HPP' ;
TAB1 . 'PROCEDURE_CHARMECA' = VRAI ;
TAB1 . 'AMOI' = TABL ;
TAB1 . 'AMOI' . 'EVOL' = EV1 ;
TAB1 . 'AMOI' . 'POINT' = P2 ;
```

Programme principal

EXERCICE 1 : POUTRE AVEC FORCE SUIVEUSE

■ Solution (ter)

- idem mais la force est re-calculée sur la configuration déformée à la fin du pas de temps !
- **plus précis** : une fois convergé, la configuration déformée est exactement perpendiculaire à la force appliquée
- **attention : peut devenir instable** 😱 !

```
DEBP CHARMECA T1*'TABLE' TPS1*'FLOTTANT' ;
```

```
* donnees utiles
```

```
TAM = T1 . 'AMOI' ;
```

```
PF = TAM . 'POINT' ;
```

```
EV = TAM . 'EVOL' ;
```

```
* angle de rotation de la section droite
```

```
DRS = 0. ;
```

```
SI (EGA (TYPE ZDEPT) 'CHPOINT') ;
```

```
DRS = EXTR ZDEPT 'RZ' PF ;
```

```
FINSI ;
```

```
U0 = T1 . 'CONTINUATION' . 'DEPLACEMENTS' ;
```

```
RS = (EXTR U0 'RZ' PF) + DRS ;
```

```
RSD = RS * 180. / PI ;
```

```
* direction de l'effort
```

```
DIR1 = (COS RSD) (SIN RSD) ;
```

```
* creation du second membre
```

```
VAL1 = IPOL TPS1 EV ;
```

```
F1 = FORC (VAL1 * DIR1) PF ;
```

```
T2 = TABLE ;
```

```
T2 . 'ADDI_SECOND' = F1 ;
```

```
FINP T2 ;
```

```
***TAB1 . 'CHARGEMENT' = CHAR1 ;
```

```
TAB1 . 'GRANDS_DEPLACEMENTS' = VRAI ;
```

```
TAB1 . 'PREDICTEUR' = 'HPP' ;
```

```
TAB1 . 'PROCEDURE_CHARMECA' = VRAI ;
```

```
TAB1 . 'AMOI' = TABL ;
```

```
TAB1 . 'AMOI' . 'EVOL' = EV1 ;
```

```
TAB1 . 'AMOI' . 'POINT' = P2 ;
```



EXERCICE 2 : RUPTURE PAR SUPPRESSION D'ÉLÉMENTS

TÉLÉCHARGEZ LE FICHIER INITIAL SUR LE SITE WEB :

[HTTP://WWW-CAST3M.CEA.FR/INDEX.PHP?PAGE=EXEMPLES&EXEMPLE=FORMATION_PASAPAS_2_INITIAL](http://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_2_initial)

EXERCICE 2 : RUPTURE PAR SUPPRESSION D'ÉLÉMENTS

■ Traction d'une plaque perforée élasticité, grands déplacements



Contrainte σ_1 et
forces de réaction
sur la déformée

■ Objectif : modéliser la rupture en supprimant les éléments au cours du calcul.

*On utilisera un critère simple sur la 1^{ère} contrainte principale :
rupture si $\sigma_1 \geq 22 \text{ GPa}$*

À vous de jouer !

Quelques indices ...

■ Opérateurs utiles

PRIN : calcul des contraintes principales

CHAN : changement des points support d'un champ par éléments

ELEM : isoler les éléments où un champ par élément vérifie un critère

REDU : réduction d'un modèle sur un sous maillage

■ Informations utiles

Modifier les **objets temporaires** du calcul dans **TAB1 . WTABLE**

■ Solution

- raffiner les pas de temps
- utiliser la procédure **PERSO1**
- extraire le modèle et les contraintes calculées (T1 . **ESTIMATION**)
- calculer les contraintes principales
- déterminer le maillage des « éléments non rompus »
- réduire le modèle sur ce maillage
- écraser le modèle dans la table de travail **WTABLE**

```
DEBP PERSO1 T1*'TABLE' ;
MO1 = T1 . 'WTABLE' . 'MODELE' ;
SG1 = T1 . 'ESTIMATION' . 'CONTRAINTES' ;
SGP1 = PRIN SG1 MO1 ;
SG11 = CHAN 'GRAVITE' (EXCO 'SI11' SGP1) MO1 ;
MAIL2 = SG11 ELEM 'INFERIEUR' 2.2E10 ;
MO2 = REDU MO1 MAIL2 ;
T1 . 'WTABLE' . 'MODELE' = MO2 ;
FINP ;
```


Procédure **PERSO1**

```
TAB1 . 'PROCEDURE_PERSO1' = VRAI ;
TAB1 . 'TEMPS_CALCULES' = PROG 0.1 'PAS' 0.1 0.6
                                'PAS' 0.01 1. ;
```

Programme principal

EXERCICE 2 : RUPTURE PAR SUPPRESSION D'ÉLÉMENTS

■ Résultats



Modèle peu robuste :
résultat très sensible à la discrétisation espace/temps

■ Solution (bis)

- raffiner les pas de temps
- utiliser la procédure **PERSO1**
- extraire le modèle et les contraintes calculées (T1 . **ESTIMATION**)
- calculer les contraintes principales
- déterminer le maillage des éléments non rompus
- réduire les **blocages** sur ce maillage
- écraser les blocages dans la table de travail **WTABLE**

```

DEBP PERSO1 T1*'TABLE' ;
M01 = T1 . 'WTABLE' . 'MODELE' ;
MAIL1 = EXTR M01 'MAIL' ;
SG1 = T1 . 'ESTIMATION' . 'CONTRAINTES' ;
SGP1 = PRIN SG1 M01 ;
SG11 = CHAN 'GRAVITE' (EXCO 'SI11' SGP1) M01 ;
MAIL2 = SG11 ELEM 'INFERIEUR' 2.2E10 ;
MAIL3 = DIFF MAIL1 MAIL2 ;
NE3 = NBEL MAIL3 ;
SI (NE3 > 0) ;
  MESS ' [PERSO1 :] Suppression de' NE3 'elements' ;
  BL01 = T1 . 'WTABLE' . 'BLOCAGES_MECANIQUES' ;
  MAILBL1 = EXTR BL01 'MAIL' ;
  MAILBL2 = MAILBL1 ELEM 'APPU' 'LARG' MAIL3 ;
  MAILBL3 = DIFF MAILBL1 MAILBL2 ;
  BL02 = REDU BL01 MAILBL3 ;
  T1 . 'WTABLE' . 'BLOCAGES_MECANIQUES' = BL02 ;
FINSI ;
FINP ;

```

Procédure **PERSO1**

```


TAB1 . 'PROCEDURE_PERSO1' = VRAI ;
TAB1 . 'TEMPS_CALCULES'   = PROG 0.1 'PAS' 0.1 0.6
                          'PAS' 0.01 1. ;

```

Programme principal PAGE 44

EXERCICE 2 : RUPTURE PAR SUPPRESSION D'ÉLÉMENTS

Results



Modèle peu robuste :
résultat très sensible à la discrétisation espace/temps

FONCTIONNEMENT DE TRANSNON SOLVEUR THERMIQUE

■ Equations locale de la thermique transitoire



densité de flux de chaleur

$$\vec{\varphi} = -\lambda \overrightarrow{\text{grad}}(T)$$

sur V

équation de la chaleur

$$\rho c_p \frac{\partial T}{\partial t} + \text{div } \vec{\varphi} - q = 0$$

sur V

flux imposé

$$\vec{\varphi} \cdot \vec{n} = \varphi_{imp} + \underbrace{h(T_f - T)}_{\text{convection}} + \underbrace{\varepsilon \sigma (T_\infty^4 - T^4)}_{\text{rayonnement}}$$

sur ∂V^φ

température imposée

$$T = T_{imp}$$

sur ∂V^T

■ Formulation faible + discrétisation EF

$$[C]\{\dot{T}\} + [K]\{T\} = \{F\}$$

Vecteur des chaleurs nodales équivalentes (W) $\{F\} = \{F\}^\varphi + \{F\}^V$

$$\{F\}^\varphi = \int_{\partial V \varphi} [N]^T \varphi_{imp} dS \quad \text{des flux imposés } \varphi_{imp} \text{ (J.m}^{-2}\text{)} \quad \text{(FLUX)}$$

$$+ \int_{\partial V \varphi} [N]^T h T_f dS \quad \text{des flux de convection imposés } h T_f \text{ (J.m}^{-2}\text{)} \quad \text{(CONV)}$$

$$+ \int_{\partial V \varphi} [N]^T \varepsilon \sigma (T_\infty^4 - T^4) dS \quad \text{des flux de rayonnement imposés (J.m}^{-2}\text{)}$$

$$\{F\}^V = \int_V [N]^T q dV \quad \text{des sources volumiques imposées } q \text{ (J.m}^{-3}\text{)} \quad \text{(SOUR)}$$

Matrices

$$[C] = \int_V \rho c_p [N]^T [N] dV \quad \text{matrice de capacité (J.K}^{-1}\text{)} \quad \text{(CAPA)}$$

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V \varphi} h [N]^T [N] dS \quad \text{matrice de conductivité (W.K}^{-1}\text{)} \quad \text{(COND)}$$

$[N]$ matrice des fonctions de forme (d'interpolation)

$[B]$ matrice des dérivées des fonctions de forme

■ Formulation faible + discrétisation EF

$$[C]\{\dot{T}\} + [K]\{T\} = \{F\}$$

■ Discrétisation temporelle : **thêta méthode**

$$[C]_* \frac{\{T\}_{n+1} - \{T\}_n}{\Delta t} + [K]_* \{T\}_* = \{F\}_*$$

avec :

Δt pas de temps

$\{T\}_n$ vecteur des températures nodales au temps t_n

$\{T\}_{n+1}$ vecteur des températures nodales au temps t_{n+1}

$[C]_*$, $[K]_*$ et $\{F\}_*$ sont estimées :

au temps $t_* = \theta t_{n+1} + (1 - \theta)t_n$


à la température $\{T\}_* = \theta \{T\}_{n+1} + (1 - \theta)\{T\}_n$

θ coefficient de relaxation (entre 0 et 1)

$\theta = 0$ schéma explicite

$\theta = 1$ schéma implicite (valeur par défaut)

FONCTIONNEMENT DE TRANSNON



Récupération des options choisies, instants de calcul, unités (°C ou K), état initial, calcul des facteurs de forme...

Mise à jour des variables externes, caractéristiques matériaux
Calcul matrice de conductivité $[K]_*$ (cond. + conv. + rayo. + bloc.)
et de capacité $[C]_*$ à la température $\{T\}_n$ et au temps t_*
Calcul du second membre $\{F\}_*$ (T impo. + T conv. + T rayo. + Flux impo.)
à la température $\{T\}_n$ et au temps t_*

On définit alors l'opérateur : $[L]_* = \frac{1}{\Delta t} [C]_* + \theta [K]_*$

Calcul de l'incrément de température sur le pas : $\Delta\{T\}^{i+1} = [L]_*^{-1}\{F\}_*$
on en déduit : $\{T\}_{n+1}^{i+1} = \{T\}_n + \Delta\{T\}^{i+1}$
et aussi : $\{T\}_*^{i+1} = \theta\{T\}_{n+1}^{i+1} + (1 - \theta)\{T\}_n$

Mesure de la variation maxi. de l'incrément de température
entre deux itérations successives :
Crit = [MAXI 'ABS' $(\Delta\{T\}^{i+1} - \Delta\{T\}^i)$] / [MAXI 'ABS' $\{T\}_{n+1}^{i+1}$]

Si le critère est suffisamment petit
Crit \leq 'PRECISION' (1.E-4 par défaut)

Procédure utilisateur

Remarque :
lorsque TRANSNON est utilisée par PASAPAS,
la **boucle sur les pas de temps** n'est effectuée
qu'une seule fois

EXERCICE 3 :
SOURCE DE CHALEUR DÉPENDANTE DE LA
TEMPÉRATURE


TÉLÉCHARGEZ LE FICHER INITIAL SUR LE SITE WEB :

[HTTP://WWW-CAST3M.CEA.FR/INDEX.PHP?PAGE=EXEMPLES&EXEMPLE=FORMATION_PASAPAS_3_INITIAL](http://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_3_initial)


EXERCICE 3 : SOURCE DE CHALEUR DÉPENDANTE DE LA TEMPÉRATURE

■ Section carrée chauffée par une source et refroidie par convection


Convection



Convection



Temperature au temps 0.00000E+00 (0%)




Temperature aux coins vs Temps

- **Objectif : rendre le problème variable !**
 - Conductivité fonction de la température
 - Convection fonction du temps
 - Source fonction de la température

$$\lambda(T) = 0,3 \cdot T + 200$$

$$f(T) = 4\,000\,000 \cdot e^{-\left(\frac{T-1000}{700}\right)^2}$$



À vous de jouer !

Quelques indices ...

■ Objets utiles

MAIL1 : maillage de la source

MOD1 : modèle thermique réduit sur MAIL1

■ Opérateurs utiles

REDU : pour réduire le champ de températures sur la zone "source"

SOUR : pour imposer une source volumique de chaleur

■ Solution

- utiliser la procédure **PERSO2** (appelée après chaque pas)
- re-calculer la source (2nd membre) selon la température au début du pas
- en faire un chargement et écraser celui dans **WTABLE**

```
DEBP PERSO2 T1*'TABLE' ;
  MA1 = T1 . 'AMOI' . 'MAIL' ;
  MO1 = T1 . 'AMOI' . 'MODE' ;
* reduction du champ de temperatures au maillage de la source
  CHT1 = T1 . 'ESTIMATION' . 'TEMPERATURES' ;
  CHT2 = REDU CHT1 MA1 ;
* calcul du champ de source a partir du champ de temperature
  CHS = 4.E6 * (EXP (-1. * (((CHT2 - 1000.) / 700.) ** 2))) ;
* creation du second membre et d'un chargement
  CHFL1 = SOUR MO1 CHS ;
  EV1 = EVOL 'MANU' (PROG 0. TPSMAX) (PROG 1. 1.) ;
  CHA1 = CHAR 'Q' CHFL1 EV1 ;
* modification du chargement dans la table de calcul
  T1 . 'WTABLE' . 'CHARGEMENT' = CHA1 ;
FINP T2 ;
```

Procédure **PERSO2**

```
LT = PROG -1.E6 1.E6 ;
LL = 200. + (0.3 * LT) ;
LAMBDA = EVOL 'MANU' 'T' LT 'K' LL ;

HCONV = EVOL 'MANU' 'TEMP' (PROG 0. TPSMAX) 'H' (PROG 10. 200.) ;

VSOUR = 4.E6 * (EXP (-1. * ((T_INI - 1000.) / 700.) ** 2)) ;

TAB1 . 'PROCEDURE_PERSO2' = VRAI ;
TAB1 . 'AMOI' = TABL ;
TAB1 . 'AMOI' . 'MAIL' = MAIL1 ;
TAB1 . 'AMOI' . 'MODE' = MOD1 ;
```

Programme principal

EXERCICE 3 : SOURCE DE CHALEUR DÉPENDANTE DE LA TEMPÉRATURE

■ Solution (bis)

- supprimer le chargement CHA1 (source de chaleur initiale)
- utiliser la procédure **CHARTHER** (appelée à chaque itération d'un pas)
- re-calculer la source (2nd membre) selon la température au début du pas

```
DEBP CHARTHER T1*'TABLE' TPS1*'FLOTTANT' ;
  MA1 = T1 . 'AMOI' . 'MAIL' ;
  MO1 = T1 . 'AMOI' . 'MODE' ;
* reduction du champ de temperatures au maillage de la source
  CHT1 = T1 . 'ESTIMATION' . 'TEMPERATURES' ;
  CHT2 = REDU CHT1 MA1 ;
* calcul du champ de source a partir du champ de temperature
  CHS = 4.E6 * (EXP (-1. * (((CHT2 - 1000.) / 700.) ** 2))) ;
* creation du second membre
  CHFL1 = SOUR MO1 CHS ;
* sortie du second membre
  T2 = TABL ;
  T2 . 'ADDI_SECOND' = CHFL1 ;
FINP T2 ;
```

Procédure CHARTHER

```
LT = PROG -1.E6 1.E6 ;
LL = 200. + (0.3 * LT) ;
LAMBDA = EVOL 'MANU' 'T' LT 'K' LL ;

HCONV = EVOL 'MANU' 'TEMP' (PROG 0. TPSMAX) 'H' (PROG 10. 200.) ;

***TAB1 . 'CHARGEMENT' = CHA1 ;
TAB1 . 'PROCEDURE_CHARTHER' = VRAI ;
TAB1 . 'AMOI' = TABL ;
TAB1 . 'AMOI' . 'MAIL' = MAIL1 ;
TAB1 . 'AMOI' . 'MODE' = MOD1 ;
```

Programme principal

■ Solution (ter)

- idem
- utiliser la procédure **CHARTHER** (appelée à chaque itération d'un pas)
- re-calculer la source (2nd membre) selon la température à l'itération précédente (i.e. à la fin du pas)

```
DEBP CHARTHER T1*'TABLE' TPS1*'FLOTTANT' ;
  MA1 = T1 . 'AMOI' . 'MAIL' ;
  MO1 = T1 . 'AMOI' . 'MODE' ;
* reduction du champ de temperatures au maillage de la source
  CHT1 = T1 . 'WTABLE' . 'THER_COURANT' ;
  CHT2 = REDU CHT1 MA1 ;
* calcul du champ de source a partir du champ de temperature
  CHS = 4.E6 * (EXP (-1. * (((CHT2 - 1000.) / 700.) ** 2))) ;
* creation du second membre
  CHFL1 = SOUR MO1 CHS ;
* sortie du second membre
  T2 = TABL ;
  T2 . 'ADDI_SECOND' = CHFL1 ;
FINP T2 ;
```

Procédure **CHARTHER**

```
LT = PROG -1.E6 1.E6 ;
LL = 200. + (0.3 * LT) ;
LAMBDA = EVOL 'MANU' 'T' LT 'K' LL ;


HCONV = EVOL 'MANU' 'TEMP' (PROG 0. TPSMAX) 'H' (PROG 10. 200.) ;

***TAB1 . 'CHARGEMENT' = CHA1 ;
TAB1 . 'PROCEDURE_CHARTHER' = VRAI ;
TAB1 . 'AMOI' = TABL ;
TAB1 . 'AMOI' . 'MAIL' = MAIL1 ;
TAB1 . 'AMOI' . 'MODE' = MOD1 ;
```




Programme principal

EXERCICE 3 : SOURCE DE CHALEUR DÉPENDANTE DE LA TEMPÉRATURE

Résultats



Temperature au temps 0.00000E+00 (0%)




EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU

TÉLÉCHARGEZ LE FICHIER INITIAL SUR LE SITE WEB :

[HTTP://WWW-CAST3M.CEA.FR/INDEX.PHP?PAGE=EXEMPLES&EXEMPLE=FORMATION_PASAPAS_4_INITIAL](http://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_4_initial)

EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU

- Deux barreaux se dilatent et entrent en contact
dilatation thermique
contact mécanique unilatéral



- **Objectif : ajouter le contact thermique**, c'est-à-dire le transfert de chaleur lorsque le contact mécanique a lieu.

À vous de jouer !

Quelques indices ...

■ Objets utiles

P2 et P3 : points à gauche et droite du jeu

L2 et L3 : lignes à gauche et droite du jeu

■ Opérateurs utiles

COOR : coordonnées de points

RELA : relation entre degrés de liberté

■ Modifier les blocages thermiques de **WTABLE**

EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU

■ Solution

- vérifier la **convergence mécanique + thermique**
- créer un second blocage pour imposer une relation :
T de la ligne L2 = T de la ligne L3 avec **RELA**
- utiliser **REEV_MEC** pour modifier les conditions aux limites thermiques
- calculer le jeu à l'instant de l'itération
- écraser les blocages thermiques dans **WTABLE** selon la valeur du jeu

```
DEBP REEV_MEC T1*'TABLE' N1*'ENTIER' ;
U1 = T1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
WT = T1 . 'WTABLE' ;
TAM = T1 . 'AMOI' ;
* calcul du jeu
PT2 = TAM . 'POINT_2' ;
PT3 = TAM . 'POINT_3' ;
X2 = (COOR 1 PT2) + (EXTR U1 'UX' PT2) ;
X3 = (COOR 1 PT3) + (EXTR U1 'UX' PT3) ;
J1 = X3 - X2 ;
* si jeu ferme, on utilise le blocage thermique initial + le RELA
SI (J1 <EG 1.E-15) ;
    WT . 'BLOCAGES_THERMIQUES' = (TAM . 'BLOQ_0') ET
    (TAM . 'BLOQ_1') ;
* si jeu ouvert, on utilise le blocage thermique initial seul
SINON ;
    WT . 'BLOCAGES_THERMIQUES' = TAM . 'BLOQ_0' ;
FINSI ;
FINP ;
```

Procédure **REEV_MEC**

```
TAB1 . 'CONVERGENCE_MEC_THE' = VRAI ;
TAB1 . 'PROCEDURE_REEV_MEC' = VRAI ;
TAB1 . 'AMOI' = TABL ;
TAB1 . 'AMOI' . 'POINT_2' = P2 ;
TAB1 . 'AMOI' . 'POINT_3' = P3 ;
TAB1 . 'AMOI' . 'BLOQ_0' = CL_TH ;
TAB1 . 'AMOI' . 'BLOQ_1' = RELA 'T' L2 - 'T' L3 ;
```

Programme principal

EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU



T

Résultats

Absence de contact : les barreaux sont isolés




Lors du contact : conduction "parfaite"




EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU

■ Résultats



Évolution temporelle du jeu

Évolution temporelle de la température
à gauche et à droite du jeu

- Les modifications effectuées dans **REEV_MEC** sont prises en compte sur le même pas de temps (grâce à la boucle de convergence thermo-mécanique **BO_BOTH**)
- Ce n'est pas le cas de la procédure **PERSO1**, où les modifications sont prises en compte sur le pas de temps suivant

EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU

■ Solution (bis)

- transfert par **convection** dans un élément de raccord
- utilisation d'un **coef. d'échange variable** selon la pression de contact
- pression de contact décrite par un chargement 'PC'
- mise à jour du chargement 'PC' dans **REEV_MEC**

```

DEBP REEV_MEC T1*'TABLE' N1*'ENTIER' ;
R1 = T1 . 'ESTIMATION' . 'REACTIONS' ;
WT = T1 . 'WTABLE' ;
TAM = T1 . 'AMOI' ;
L3 = TAM . 'LIGNE_3' ;
MR = TAM . 'MAIL' ;
SI ((T1 . 'ESTIMATION' . 'TEMPS') NEG 0.) ;
* calcul de la pression de contact
  PARA_P = (MAXI 'ABS' (RESU (REDU R1 L3))) / (MESU L3) ;
* chargement décrivant le paramètre 'PC'
  CH_PC = MANU 'CHPO' MR 'PC' PARA_P ;
  EV1 = TAM . 'EVOL' ;
  CHA3 = CHAR 'PC' CH_PC EV1 ;
* on écrase le chargement global de WTABLE
  WT . 'CHARGEMENT' = (TAM . 'CHAR_0') ET CHA3 ;
FINS ;
FINP ;

```

```

** Modele de convection variable entre les barreaux
MRACC = RACC (1.1 * JEU_INI) L2 L3 ;
MODRACC = MODE MRACC 'THERMIQUE' 'CONVECTION' ;
HVSP = EVOL 'MANU' 'PC' (PROG 0. 5.E8 5.1E8)
      'H' (PROG 0. 1.E4 1.E4) ;
MATRACC = MATE MODRACC 'H' HVSP ;
MOD_TH = MODT1 ET MODT2 ET MODRACC ;
MAT_TH = MATT1 ET MATT2 ET MATRACC ;


** Chargement initial décrivant la pression de contact
CH_PC = MANU 'CHPO' MRACC 'PC' 0. ;
CHAPC = CHAR 'PC' CH_PC EV1 ;

TAB1 . 'CHARGEMENT'           = CHA1 ET CHA2 ET CHAPC ;
TAB1 . 'CONVERGENCE_MEC_THE' = VRAI ;
TAB1 . 'PROCEDURE_REEV_MEC'  = VRAI ;
TAB1 . 'AMOI'                 = TABL ;
TAB1 . 'AMOI' . 'LIGNE_3'     = L3 ;
TAB1 . 'AMOI' . 'MAIL'        = MRACC ;
TAB1 . 'AMOI' . 'EVOL'        = EV1 ;
TAB1 . 'AMOI' . 'CHAR_0'      = CHA1 ET CHA2 ;


```


EXERCICE 4 : FERMETURE THERMO-MÉCANIQUE D'UN JEU

■ Résultats (bis)



Évolution temporelle du jeu

Évolution temporelle de la température
à gauche et à droite du jeu

- Le transfert de chaleur se fait à des températures différentes à l'interface
- Les variations de température sont adoucies

ANNEXES

■ Généralités

NB_BOTH (ENTIER)

Nombre max. d'itérations de la boucle de convergence thermo-mécanique

MAXITERATION (ENTIER)

Nombre max. d'itérations (49)

■ Mécanique

PRECISION (FLOTTANT)

Critère pour comparer le résidu (1.E-4)

FTOL (FLOTTANT)

Tolérance pour l'équilibre des efforts

MTOL (FLOTTANT)

Tolérance pour l'équilibre des moments

GRANDS_DEPLACEMENTS (LOGIQUE)

Configuration de référence = configuration déformée

PREDICTEUR = 'HPP'

Hypothèse petits déplacements → 1^{ère} convergence puis passage sous

l'hypothèse grands déplacements → 2^{nde} convergence

PRECISINTER (FLOTTANT)

Précision pour le problème local d'intégration des lois constitutives (1.E-8)

CONVERGENCE_FORCEE (LOGIQUE)

Utilisation ou non de la convergence forcée en cas de non convergence (VRAI)

MAXSOUSPAS (ENTIER)

Nombre max. de sous-pas en convergence forcée (200)

DELTAITER (ENTIER)

Nombre de pas d'écart pour test de non convergence

■ Thermique

PROCEDURE_THERMIQUE (MOT)

Procédure de calcul à utiliser :

- **NONLINEAIRE** *procédure TRANSNON*

- **LINEAIRE** *procédure TRANSLIN*

- **DUPONT** *procédure DUPONT2*

RELAXATION_THETA (FLOTTANT)

Coefficient de relaxation pour la θ -méthode (1)

■ Couplage thermo-mécanique

CONVERGENCE_MEC_THE (LOGIQUE)

*Indique que l'on souhaite ré-itérer la boucle de calcul thermo-mécanique en cas de dépendance mutuelle (**FAUX**)*

CRITERE_COHERENCE (FLOTTANT)

*Précision pour la convergence thermo-mécanique, test sur la thermique (**1.E-2**)*

PROJECTION (LOGIQUE)

*Indique que le problème est couplé mais que les maillages en mécanique et en thermique sont différents (**FAUX**)*

■ Algorithme UNPAS simplifié (voir la procédure @SOLVMEC) [petits déplacements]

```

fex1 = TIRE char tps1 'MECA' ;
flx1 = TIRE char tps1 'DIMP' ;
resi0 = fex1 + fr0 - fi0 + flx1 - flx0 ;
fref = MAXI 'ABS' (fex1 + fr0) ;
k = (RIGI mod mat) ET blo ;
REPE b1 100 ;
  ddu = RESO k resi0 ;
  u1 = u0 + ddu ;
  fr1 = REAC blo u1 ;
  eps1 = EPSI mod u1 ;
  sig1 = ... COMP eps0 eps1 ...
  fi1 = BSIG mod sig1 ;
  resi1 = fex1 + fr1 - fi1 ;
  crit1 = (MAXI 'ABS' resi1) / fref ;
  SI (crit1 < precis) ;
    QUIT b1 ;
  FINSI ;
  u0 = u1 ;
  resi0 = resi1 ;
FIN b1 ;

```

■ Algorithme UNPAS simplifié (voir la procédure @SOLVMEC) [grands déplacements, lagrangien réactualisé]

```
fex1 = TIRE char tps1 'MECA' ;
flx1 = TIRE char tps1 'DIMP' ;
resi0 = fex1 + fr0 - fi0 + flx1 - flx0 ;
fref = MAXI 'ABS' (fex1 + fr0) ;
```

```
conf0 = FORM u0 ;
u00 = u0 ;
k = (RIGI mod mat) ET blo ET (KSIG mod sig0) ;
```

```
REPE b1 100 ;
  ddu = RESO k resi0 ;
  u1 = u0 + ddu ;
  fr1 = REAC blo u1 ;
```

```
  du = u1 - u00 ;
  deps = EPSI mod du ;
  eps1 = eps0 + deps ;
  sig1 = ... (COMP eps0 eps1, EXCO, ...)
```

```
  sig1 = PICA mod sig1 du ;
  FORM u1 ;
  fi1 = BSIG mod sig1 ;
  FORM conf0 ;
```

```
  resi1 = fex1 + fr1 - fi1 ;
  crit1 = (MAXI 'ABS' resi1) / fref ;
  SI (crit1 < precis) ;
    QUIT b1 ;
```

```
  FINSI ;
  u0 = u1 ;
  resi0 = resi1 ;
```

```
FIN b1 ;
```

on se place en configuration courante (début du pas de temps)

incrément de déformation

transport des contraintes sur configuration fin du pas de temps

intégration sur configuration fin du pas de temps

■ Normes de convergence (après le 1er RESO)

$$F^{ref} = \frac{|\delta\{U\}_1^1 \cdot \{F\}_1^{S+V} - \{\lambda\}_1^1 \cdot (\{d\}_1 - [A]^T \cdot \{U\}_1^0)|}{\max_{ddl\ depl} |\delta\{U\}_1^1| + xpetit} + \frac{\max_{ddl\ depl} |\delta\{U\}_1^1| \cdot \max|[B]\{\sigma\}_1^0|}{\max_{ddl\ depl} |\delta\{U\}_1^1| + xpetit} + \max|[B]\{\sigma\}_1^0|$$

$$M^{ref} = \frac{|\delta\{U\}_1^1 \cdot \{F\}_1^{S+V} - \{\lambda\}_1^1 \cdot (\{d\}_1 - [A]^T \cdot \{U\}_1^0)|}{\max_{ddl\ rota} |\delta\{U\}_1^1| + xpetit} + \frac{\max_{ddl\ depl} |\delta\{U\}_1^1| \cdot \max|[B]\{\sigma\}_1^0|}{\max_{ddl\ rota} |\delta\{U\}_1^1| + xpetit} + xpetit$$

■ Mesures du résidu (à chaque itération de UNPAS)

$$XCONV = \frac{\max\left\{ \max_{ddl\ depl} |\{F\}_1^{S+V} - [A]^T \cdot \{\lambda\}_1^{i+1} - [B]\{\sigma\}_1^{i+1}| ; \max|[A]^T \cdot (\{\lambda\}_1^{i+1} - \{\lambda\}_1^i)| \right\}}{F^{ref}}$$

$$XCONVM = \frac{\max_{ddl\ rota} |\{F\}_1^{S+V} - [A]^T \cdot \{\lambda\}_1^{i+1} - [B]\{\sigma\}_1^{i+1}|}{M^{ref}}$$

- Mesure de la variation d'incrément de déformation entre 2 itérations

$$\text{DEPSTDM} = \max|\Delta\{\varepsilon\}_1^{i+1} - \Delta\{\varepsilon\}_1^i|$$

- La convergence est atteinte si

$$\begin{aligned} \text{XCONV} &< \text{PRECISION} \\ \text{XCONVM} &< \text{PRECISION} \\ \text{DEPSTDM} &< \text{PRECISION} \end{aligned}$$

- Si non convergence avérée, **on refait le pas** :
 - on redéfinit l'**état au début du pas**
 - on choisit le dernier **état "hors équilibre"** avec résidu minimal
 - on écrase ainsi, au début du pas :
 - . l'instant
 - . les propriétés matériau
 - . les déplacements
 - . les variables internes
 - . les déformations inélastiques
 - . les contraintes
 - on **recommence UNPAS avec un pas de temps de taille nulle**
 - pas d'incrément de charge, seulement le déséquilibre
- permet de trouver une solution respectant le comportement et l'équilibre mais avec un trajet de chargement différent

François DI PAOLA
Commissariat à l'énergie atomique et aux énergies alternatives
Centre de Saclay | 91191 Gif-sur-Yvette Cedex

DES/DANS
DM2S
SEMT